24p

N62-16298

NASA TN D-1453

NASA TN D-1453

TECHNICAL NOTE

D-1453

A FORTRAN II PROGRAM FOR ANALYSIS OF RADIOACTIVE DECAY CURVES

By John L. Need and Theodore E. Fessler

Lewis Research Center
Cleveland, Ohio

OTS PRICE

XEROX $ 2.60 ph

MICROFILM $ .92 mf

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

WASHINGTON October 1962

N62-16…

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

TECHNICAL NOTE D-1453

A FORTRAN II PROGRAM FOR ANALYSIS OF RADIOACTIVE DECAY CURVES

By John L. Need and Theodore E. Fessler

SUMMARY

The Fortran II language program described here was written to enable calculation of the half-lives as well as the initial magnitudes of the several components in a measured radioactive decay curve. Any of the components can be fixed as to magnitude and half-life and any of the half-lives can be fixed, provided that there be at least one variable remaining. The calculation proceeds by an iterative least-squares fit to the experimental data. Convergence is not guaranteed, and a stepwise approach may be necessary. The output consists of the amplitudes and half-lives of all the components together with the best estimate of the errors in those magnitudes calculated by the program. In addition, information is given that permits evaluation of the goodness of fit. The program as written can handle up to 10 components (including background) and up to 400 observations. It was written for an IBM 704 with a floating-point trap and an 8K memory.

INTRODUCTION

The analysis of multicomponent decay data is at best a tedious task if done by hand calculation procedures. Unless some systematic procedure such as a least-squares analysis is used, there is always the danger of systematic error due to human bias. Even when a solution is obtained, it is virtually impossible to assign an appropriate error estimate to the solution without involving further lengthy calculations. For these reasons, the problem of fitting multicomponent decay data is a natural candidate for solution by electronic high-speed computing machines. An example of a program written for the Univac for the case where the half-lives are fixed is that reported in reference 1. An excellent discussion of the general nonlinear least-squares fitting problem can be found in reference 2.

The Fortran language program described in the present report was designed to find the initial activities and unknown half-lives in a multicomponent radioactive source by a least-squares analysis of count rates. This program was developed at the Lewis Research Center for use in determining the half-life and the activation cross section of indium 117 from both Geiger-Müller counter data and multichannel γ-ray pulse-height data. The initial version did not contain any error printout or goodness of fit criteria, and experience with it led to their inclusion in the program described here. As presented, the code is useful for single- or multicomponent decay problems including those situations involving buildup of active daughter products.

It is the purpose of this report to present the mathematical method and the particulars of the program operation in sufficient detail so that others may use the code with a minimum of effort.

MATHEMATICAL METHOD

If the activity of a sample of radioactive material is measured with a counter and if it is known that each event is independent of all others, then the expected count rate may be expressed as

$$\varphi(t_k) = \sum_i A_i e^{-B_i t_k} \tag{1}$$

where φ is the count rate at the time t_k and the A_i and B_i are constants characteristic of the initial amount and the half-life of the i^{th} species in the sample, respectively. (The assumption of independence of events is equivalent to an assumption that there are no active daughter species being formed and that the counting system is perfect, i.e., no dead time, geometry changes, etc. It is shown later in this report that the program as written can be used when there is an active daughter species present.) It should be understood that equation (1) accurately describes the situation that prevails. The problem to be solved is that of determining the values of A_i and B_i that best fit an experimentally determined set of count rates φ_k determined at times t_k, where it is assumed that the expected values of φ_k are given by equation (1).

Determination of the Parameters A_i and B_i

The method used to obtain the "best fit" is the least-squares method. Thus, the problem to be solved is that of finding the set of A_i's and B_i's for which the weighted sum of the squares of the residuals

$$\sum_{k=1}^{m} \xi_k \left[\varphi(t_k) - \sum_{i=1}^{n} A_i e^{-B_i t_k} \right]^2 = R \tag{2}$$

is a minimum. Here it is assumed that the weight ξ_k of the k^{th} datum is solely dependent upon the precision with which φ_k is known, and that the value of t_k contains no uncertainty.

To make the problem amenable to solution, equation (2) is linearized and the solutions are iterated until a convergence occurs. Let

$$\left.\begin{aligned} A_i &= (1 + \alpha_i)\overline{A}_i \\ B_i &= (1 + \beta_i)\overline{B}_i \end{aligned}\right\} \tag{3}$$

where α_i and β_i are corrections applying to the approximate values $\overline{A}_i$ and $\overline{B}_i$. The linearized form of equation (2) becomes

$$\left.\begin{aligned} R &= \sum_{k=1}^{m} \xi_k \left[\varphi(t_k) - \sum_{i=1}^{n} \overline{A}_i (1 + \alpha_i) e^{-\overline{B}_i (1+\beta_i) t_k} \right]^2 \\ &\approx \sum_{k=1}^{m} \xi_k \left[\varphi(t_k) - \sum_{i=1}^{n} \overline{A}_i (1 + \alpha_i - \overline{B}_i \beta_i t_k) e^{-\overline{B}_i t_k} \right]^2 \end{aligned}\right\} \tag{4}$$

(to the first order in α_i and β_i) for m observations to be fit with n components. The problem is now that of determining the α_i and β_i such that R is a minimum for a given set of $\overline{A}_i$ and $\overline{B}_i$. After the α and β are determined, the correction terms $\overline{A}_i\alpha_i$ and $\overline{B}_i\beta_i$ are added to the original $\overline{A}_i$ and $\overline{B}_i$ to provide the input values for another determination of α_i and β_i. The calculation is terminated when all the α_i and β_i are as small as desired.

The conditions for the minimum value of R are the coupled equations

$$\left.\begin{aligned}
\frac{\partial R}{\partial \alpha_l} = 0 &= \sum_{k=1}^{m} \xi_k \left[\varphi(t_k) - \sum_{i=1}^{n} \overline{A}_i(1 + \alpha_i - \overline{B}_i\beta_i t_k)e^{-\overline{B}_i t_k}\right]\overline{A}_l e^{-\overline{B}_l t_k} \\
\frac{\partial R}{\partial \beta_l} = 0 &= \sum_{k=1}^{m} \xi_k \left[\varphi(t_k) - \sum_{i=1}^{n} \overline{A}_i(1 + \alpha_i - \overline{B}_i\beta_i t_k)e^{-\overline{B}_i t_k}\right]\overline{A}_l(-\overline{B}_l t_k)e^{-\overline{B}_l t_k}
\end{aligned}\right\} \tag{5}$$

where l assumes values from 1 to n. Rewritten, equations (5) become

$$\left.\begin{aligned}
&\sum_{k=1}^{m} \xi_k\left(\overline{A}_l e^{-\overline{B}_l t_k}\right)\left[\varphi(t_k) - \sum_{i=1}^{n} \overline{A}_i e^{-\overline{B}_i t_k}\right] \\
&\qquad = \sum_{i=1}^{n}\left[\sum_{k=1}^{m} \xi_k \overline{A}_l e^{-\overline{B}_l t_k}\left(\overline{A}_i e^{-\overline{B}_i t_k}\right)\right]\alpha_i \\
&\qquad + \sum_{i=1}^{n}\left[\sum_{k=1}^{m} \xi_k \overline{A}_l e^{-\overline{B}_l t_k}\left(-\overline{A}_i\overline{B}_i t_k e^{-\overline{B}_i t_k}\right)\right]\beta_i \\
&\text{and} \\
&\sum_{k=1}^{m} \xi_k\left(-\overline{A}_l\overline{B}_l t_k e^{-\overline{B}_l t_k}\right)\left[\varphi(t_k) - \sum_{i=1}^{n} \overline{A}_i e^{-\overline{B}_i t_k}\right] \\
&\qquad = \sum_{i=1}^{n}\left[\sum_{k=1}^{m} \xi_k\left(-\overline{A}_l\overline{B}_l t_k e^{-\overline{B}_l t_k}\right)\overline{A}_i e^{-\overline{B}_i t_k}\right]\alpha_i \\
&\qquad + \sum_{i=1}^{n}\left[\sum_{k=1}^{m} \xi_k\left(-\overline{A}_l\overline{B}_l t_k e^{-\overline{B}_l t_k}\right)\left(-\overline{A}_i\overline{B}_i t_k e^{-\overline{B}_i t_k}\right)\right]\beta_i
\end{aligned}\right\} \tag{6}$$

where the coefficients of the α_i and β_i are explicitly written out. These coupled equations can be written as

$$\sum_{k=1}^{m} \xi_k \psi_l(t_k)\left[\varphi(t_k) - \sum_{i=1}^{n} \psi_i(t_k)\right] = \sum_{i=1}^{2n} h_{l,i}\gamma_i \tag{7}$$

if the following definitions are made,

$$\left.\begin{aligned} \gamma_i &\equiv \alpha_i & 0 < i \leq n \\ \gamma_i &\equiv \beta_{i-n} & n < i \leq 2n \end{aligned}\right\} \tag{8}$$

$$\left.\begin{aligned} \psi_i(t_k) &\equiv \overline{A}_i e^{-\overline{B}_i t_k} & 0 < i \leq n \\ \psi_i(t_k) &\equiv -\overline{A}_{i-n} e^{-\overline{B}_{i-n} t_k} \overline{B}_{i-n} t_k & n < i \leq 2n \end{aligned}\right\} \tag{9}$$

and

$$\left.\begin{aligned}
h_{l,i} &\equiv \sum_{k=1}^{m} \xi_k \overline{A}_l e^{-\overline{B}_l t_k} \overline{A}_i e^{-\overline{B}_i t_k} & & l,i \leq n \\
h_{l,i} &\equiv \sum_{k=1}^{m} \xi_k \overline{A}_l e^{-\overline{B}_l t_k} \left(-\overline{A}_{i-n} \overline{B}_{i-n} t_k e^{-\overline{B}_{i-n} t_k}\right) & & l \leq n,\ n < i \leq 2n \\
h_{l,i} &\equiv \sum_{k=1}^{m} \xi_k \left(-\overline{A}_{l-n} \overline{B}_{l-n} t_k e^{-\overline{B}_{l-n} t_k}\right) \overline{A}_i e^{-\overline{B}_i t_k} & & n < l \leq 2n,\ i \leq n \\
h_{l,i} &\equiv \sum_{k=1}^{m} \xi_k \left(-\overline{A}_{l-n} \overline{B}_{l-n} t_k e^{-\overline{B}_{l-n} t_k}\right)\left(-\overline{A}_{i-n} \overline{B}_{i-n} t_k e^{-\overline{B}_{i-n} t_k}\right) & & n < l,i \leq 2n
\end{aligned}\right\} \tag{10}$$

The γ_i that satisfy equation (7) are given by

$$\gamma_i = \sum_{l=1}^{2n} (h^{-1})_{i,l} \sum_{k=1}^{m} \xi_k \psi_l(t_k) \left[\varphi(t_k) - \sum_{i=1}^{n} \psi_i(t_k) \right] \tag{11}$$

where h^{-1} is the inverse of the matrix h.

Error Analysis

Since the problem at hand treats counter data, the weight ξ_k to be applied to the k^{th} count-rate datum is determined by the actual number of counts N_k observed in the time interval τ_k from which the count rate is obtained by

$$\varphi(t_k) = \frac{N_k}{\tau_k} \tag{12}$$

The standard deviation in $\varphi(t_k)$ is $\sqrt{N_k}/\tau_k$ or, from equation (12), $\varphi(t_k)/\sqrt{N_k}$. The weights ξ_k are taken as the inverse of the square of the standard deviation of $\varphi(t_k)$ or

$$\xi_k = \frac{N_k}{\varphi^2(t_k)} = \frac{\tau_k^2}{N_k} \tag{13}$$

<u>Test using χ^2</u>. - The χ^2 test was chosen as the indicator of goodness of fit. This test determines the probability that a repetition of the observation would show greater deviations from the distribution assumed to govern the data (eq. (1) in this case). The deviations are measured by the parameter χ^2, which is defined by

$$\chi^2 \equiv \sum_{\substack{\text{all} \\ \text{data} \\ \text{points}}} \frac{(\text{observed count - expected count})^2}{\text{expected count}}$$

For a value of χ^2 equal to the number of degrees of freedom, the probability just described is about 0.5.

To apply the test, the weighted variance Q_n is used. For the present case of m data points and n components,

$$Q_n = \frac{1}{m - 2n - 1} \sum_{k=1}^{m} \xi_k \left[\varphi(t_k) - \sum_{i=1}^{n} \psi_i(t_k) \right]^2 \tag{14}$$

With this definition it can be shown (ref. 3) that, if the data are good (i.e., the assumptions leading to eq. (1) are valid), Q_n has a χ^2 distribution with $\mathcal{M} = m - 2n - 1$ degrees of freedom and its expected value approaches unity as the number of components is increased. Sometimes the minimum value obtainable from a set of data will be considerably different from 1, in which case it may be concluded that either the basic assumptions behind equation (1) do not apply or that there is some internal inconsistency in the data.

Errors in the final fit. - It is also desirable to determine the best estimates of the standard deviations σ in the final values of A_i and B_i. These errors are functions of the errors in the count-rate data only, and it can be shown from arguments in reference 4 that they are given by

$$\left.\begin{aligned} \sigma(A_i) &= \left[(h^{-1})_{i,i}Q\right]^{1/2} A_i \\ \sigma(B_i) &= \left[(h^{-1})_{i+n,i+n}Q\right]^{1/2} B_i \end{aligned}\right\} \tag{15}$$

This equation is based upon the assumptions that the weights ξ_k are the best estimates of errors in the φ_k, that the φ_k have a normal distribution about the "true" values, and that the choice of n, the number of components, is the correct one.

Reduced Number of Variables

The complete discussion so far has been carried out for the case in which there are neither fixed components nor fixed half-lives. When any component is considered as fixed, $\varphi(t_k)$ is redefined as $\varphi(t_k) - A_f e^{-B_f t_k}$; that is, the fixed component is subtracted, and the remainder is then fitted with a new value of n.

When any half-life is considered as fixed, the number of variables (and the dimension of the matrix $h_{i,l}$) is reduced by 1. In this case there are now n values of A and $n - 1$ values of B to be determined; a convenient manner in which to order the unknowns is to put the component with the fixed half-life last, so that the resultant matrix h is simply the original matrix with the last row and column removed.

DESCRIPTION OF THE PROGRAM

Input Data

The input consists of one card that serves as a title and then two cards that carry the number of components N, the number of data points M, the number of components fixed in both magnitude and half-life NC, and the number of components fixed in half-life NE. The next M cards carry the values of count rate, time, and weight for each data point. There follow N cards with the initial values for the magnitudes and half-lives and finally one card with the convergence requirement and a weight factor. This completes the data for one case.

If fixed components or fixed half-lives are desired, the input values of A_i and B_i must be ordered such that components with both A_i and B_i variable come first, components with B_i fixed second, and last of all, components with both A_i and B_i fixed.

Sample data used to test the program are shown in table I as they appear on the input data cards. These data were created by establishment of a two-component decay plus a background corresponding to equation (1) and use of a table of random numbers to distribute the count-rate "data" about the expected values with a Poisson distribution. For convenience the input and output are expressed in terms of half-life rather than decay constant, the conversion

$$B_i = -\frac{0.693}{(t_{1/2})_i}$$

being accomplished within the program. Thus the "data" of table I represent the case in which the "true" constants are

$$A_1 = 103.97 \qquad (t_{1/2})_1 = 27.73 \text{ min}$$

$$A_2 = 75.00 \qquad (t_{1/2})_2 = 103.97 \text{ min}$$

$$A_3 = 12.50 \qquad (t_{1/2})_3 = \infty \text{ (background)}$$

and in which all but $(t_{1/2})_3$ are to be found; thus,

$$N = 3 \qquad M = 20$$

$$NE = 1 \qquad NC = 0$$

for this case. Following the 20 data points are the three cards with the trial values of $\overline{A}_i$ and $\overline{B}_i$ and the card with the convergence requirement and a weight factor. The convergence requirement is the maximum value α_i and β_i may have at the completion of the calculation. Both half-lives and count times must be in the same units.

Three options are available for the input weight values:

(1) If the weights are to be those of equation (12), the input weight value should be the number of counts in the datum N_k. The program computes the weight

$$\xi_k = \frac{N_k'}{\varphi_k^2} W$$

where W is the constant weight factor appearing on the last data card. This factor is used when it is convenient to enter the total number of counts as a scaled number N_k', as is often the case with counter data, and is chosen such that $N_k = N_k'W$.

(2) If all points are to be weighted equally, the input weight is entered as zero and the program sets

$$\xi_k = \frac{W}{\varphi_k}$$

(3) If any other weighting is desired, the program is signalled by the entrance of the input weight as the negative of the desired weight, and the program multiplies this value by -1 to obtain the weight used.

Program Operation

The program first prints a general heading to identify the calculation. (A listing of the program is given in the appendix.) The input data are then read for one case, and the first three cards are printed to identify the case. The weights are calculated, and then indices to govern the flow of the program. These indices give the size of the matrix, limits on summations, and so forth. The input values of $\overline{A}_i$ and $\overline{B}_i$ are printed and then the main part of the computation is performed.

The program computes the values of the various functions in expression (7) and does appropriate bookkeeping when there are fixed components and fixed half-lives. The matrix is inverted, and the values of the correction terms γ_i are calculated. They are compared with the convergence requirement. If all the γ_i are less than the convergence requirement, the program calculates the final values of $\overline{A}_i$ and $\overline{B}_i$ and proceeds with the output. If at least one γ_i is larger than the convergence requirement, the program calculates the correction terms $\alpha_i \overline{A}_i$ and $\beta_i \overline{B}_i$ and adds them to the original input values to obtain a new set of $\overline{A}_i$ and $\overline{B}_i$ and begins anew with the print-out of the new set of $\overline{A}_i$ and $\overline{B}_i$. Generally no more than 10 iterations are required to obtain convergence with a convergence requirement of 0.001.

If convergence is not achieved, this will be evidenced either by a failure of the matrix inversion routine or by some act that triggers the floating-point trap. Both of these events return the computation to the beginning of the next case. Since convergence to positive values of $\overline{A}_i$ and $\overline{B}_i$ can be obtained by going through some negative values, the occurrence of negative values for the $\overline{A}_i$ and $\overline{B}_i$ does not stop the program.

The output consists of the final values of $\overline{A}_i$ and $\overline{B}_i$, the values of the last set of γ_i, and the values of $\sigma(A_i)$ and $\sigma(B_i)$ (the best estimates of the standard deviation) for those values of A_i and B_i determined by the program. Both the weighted variance Q_n and the sum of the weighted squared residuals are printed out. Finally, the time, input counting rate, weight used, calculated counting rate, percent deviation, and weighted squared residual are given for each data point.

The results of using the program to solve the sample data of table I are shown in table II, which contains the program output for these "data." The test using χ^2 tells that the data are statistically good, and the $\sigma(A)_i$ and $\sigma(B)_i$ values agree quite well with the differences between the calculated values of A_i and B_i and the true values from which the "data" were derived.

Inversion of the matrix $|h|_{k,l}$ to obtain $|h^{-1}|_{k,l}$ is accomplished in a subroutine named MATINV (X,M). In this subroutine, a matrix X of rank M is inverted by means of a unit matrix of the same order as in the procedure given in reference 5. This subroutine is generalized to the extent that, if a zero is encountered in the first element of a pivotal row, the matrix is reordered by rows to bring a nonzero element to the pivotal position, and the reordering is recorded so that the inverse obtained may be unscrambled by columns to arrive at the correct inverse of X.

The subroutine MATINV (X,M) is provided with an error exit if either the value of the argument M is improper (less than 1) or the inverse is nonexistent. In either case sense light 1 is turned on to signal the calling program of trouble.

A second subroutine, TEFOUT (X,M) is used. It has the sole purpose of converting a floating-point number X to a new number $X' \times 10^M$ where $|X'|$ lies between 1.0 and 9.99 . . . and M is an integer. This form makes it possible to get an output number format of more conventional form than that provided directly by the Fortran system.

COMMENTS ON THE USE OF THE PROGRAM

The useful thing about this program is that it permits a determination of unknown half-lives and of the number of components present together with the best estimate for the errors in the derived half-lives and initial activities. Thus, the full statistical information present in the data can be abstracted, no matter how poor the data. There are three general classes of problems that this program can help to solve: (1) Problems in which the half-lives are known and there may or may not be an uncertainty in the number of components present, (2) problems in which there are one or two unknown half-lives and the number of components is known, and (3) problems in which very little is known as to the number of components and possible half-lives.

The solutions to the first two classes can be approached by the same route. The physical conditions that produced the measured activity serve as a guide to both the number and the half-lives of the possible components. A hand plot of the data yields values of the magnitudes that are used as the initial values for the machine fitting as well as some idea of the unknown half-lives if there are any. The literature will generally serve as an additional source for estimates of this half-life.

The first step in the machine calculation is the determination of the number of components present. This is done with all the half-lives held fixed with estimates used for unknown values at this point. The calculation will almost always converge in two iterations, the exceptions occurring when unrealistic fits are attempted. After the n principal components have been established (these appear in the hand plot), the question of the existence of the $(n + 1)^{st}$ component is approached as follows. Cases are run with the $(n + 1)^{st}$ component chosen successively to be each of the next three or so most probable half-lives. These fits must now be compared with the fit with n components to see whether there has been an improvement in χ^2.

In addition, the F-test should be employed to determine whether the amplitude of a particular $(n + 1)^{st}$ trial component is zero. To perform this test the statistic

$$S_{\mathcal{M}} = \frac{(m - n - 1)Q_n - (m - n - 2)Q_{n+1}}{Q_{n+1}} \qquad (16)$$

is formed, where Q_n is given by equation (14) and the subscript n refers to the number of components. Note that the denominator of equation (14) becomes $(m - n - 1)$, since each component now contributes only one variable. The F-test then states that for $S_{\mathcal{M}} \geq F_a(\mathcal{M})$ it can be assumed that A_{n+1} is nonzero with the probability a of being wrong, where $F_a(\mathcal{M})$ is defined by

$$a = \int_{F_a(\mathcal{M})}^{\infty} P_{\mathcal{M}}(X)dX$$

A table of the distribution of F_a is given in reference 6 and is reproduced here in table III for convenience. The results of this test are not always as clear-cut as might be wished, but they do provide a basis for decision. Often the addition of a component will cause the case to converge to negative values of the A_i, and (in the absence of known or suspected parent-daughter relations) this is taken as evidence that the added component is not present.

In some cases two (or more) of the trials for the $(n + 1)^{st}$ component will give fits that are difficult to choose between, and the case when both half-lives are added (as $(n + 1)^{st}$ and $(n + 2)^{nd}$ components) gives nonsense. To decide between them a run is made with $n + 1$ components but with the half-life of the $(n + 1)^{st}$ component determined by the calculation. The trial value closest to the computer-derived value is then chosen.

When the number of components has been established, any unknown half-lives are then solved for. If the final values of the unknown half-lives are widely different from the estimated values used in the determination of n, it will be useful to use these new values in a redetermination of n.

The program output should always be examined to ascertain the reliability of the results. To make this job simpler, the percent deviation (percent difference between the calculated and observed values of a count-rate datum) and the weighted square of each residual have been included in the output listing (table II). The percent deviation column may be examined for grouping of positive or negative differences. Extensive grouping of errors of one sign indicates that either a wild data point (usually near the middle of the group) is at fault or that a significant component has not been accounted for. If a wild data point is the cause, it shows up in the corresponding value of the weighted square of the residual as a high value possibly due to a data reduction error in either the count rate or the weight factor. Examination of the residuals is also of help in determining the cause of unexpectedly large values from the χ^2 test.

For the third class of problems, in which there is very little known about the number of components and the values of the half-lives, the approach described previously can be used, but it may well prove more efficient to let the half-lives vary earlier than indicated. There is no guarantee of success in these searches. When more than three half-lives are permitted to vary, the convergence gets uncertain; the program must be led very carefully to the correct region of input trial values.

There are cases where the experimenter must constrain the program. As an example, it is known that two components with half-lives of 6 and 15 minutes should be present; the program much prefers one component with a half-life of 12 minutes, but it is known that no such half-life can possibly be present. Such behavior is caused by lack of sufficient information in the count-rate data.

Negative values for the intensity of a component may arise from a real parent-daughter relation or from a systematic error in the data, such as an improper dead-time correction. Negative values may also indicate an improper assumption of the counter background. Here it should be pointed out that, for most cases, it is best to let the program determine the background amplitude. It is particularly important that the background be included in the count-rate data submitted to the machine, since prior subtraction of the background alters the statistical assumptions used in the error analysis and thus invalidates the derived quantities.

Even though the program as described seems to exclude the case of a parent-daughter relation, it does not do so. The quantities A_i obtained from the program for this case are not the initial activities, however, but are related to them in a simple manner. For a parent-daughter relation the following expressions give the time dependence of the two populations

$$\left.\begin{aligned} n_1 &= n_1^0 e^{-\lambda_1 t} \\ n_2 &= n_2^0 e^{-\lambda_2 t} + \frac{n_1^0 \lambda_1}{\lambda_2 - \lambda_1}\left(e^{-\lambda_1 t} - e^{-\lambda_2 t}\right) \end{aligned}\right\} \tag{17}$$

where subscripts 1 and 2 refer to the parent and daughter, respectively. The n^0 are the populations at time zero, and the λ are decay constants.

The detector generally has different efficiencies C_1 and C_2 for counting the decay products of the two species so that when the activity of this mixture is counted, the total count rate is given by

$$A_t = C_1\lambda_1 n_1 + C_2\lambda_2 n_2$$

$$= C_1\lambda_1 n_1^0 e^{-\lambda_1 t} + C_2\lambda_2 n_2^0 e^{-\lambda_2 t} + \frac{C_2\lambda_2 n_1^0 \lambda_1}{\lambda_2 - \lambda_1}\left(e^{-\lambda_1 t} - e^{-\lambda_2 t}\right) \qquad (18)$$

Rewriting this to make the coefficients of the two exponential terms explicit gives

$$A = C_1\lambda_1 n_1^0\left(1 + \frac{C_2}{C_1}\frac{\lambda_2}{\lambda_2 - \lambda_1}\right)e^{-\lambda_1 t} + \left(C_2\lambda_2 n_2^0 - C_1\lambda_1 n_1^0 \frac{C_2}{C_1}\frac{\lambda_2}{\lambda_2 - \lambda_1}\right)e^{-\lambda_2 t}$$

The program obtains the coefficients of the exponents so that the derived values A_1 and A_2 are

$$A_1 = C_1\lambda_1 n_1^0\left(1 + \frac{C_2}{C_1}\frac{\lambda_2}{\lambda_2 - \lambda_1}\right) = A_1^0\left(1 + \frac{C_2}{C_1}\frac{\lambda_2}{\lambda_2 - \lambda_1}\right)$$

$$A_2 = C_2\lambda_2 n_2^0 - C_1\lambda_1 n_1^0 \frac{C_2}{C_1}\frac{\lambda_2}{\lambda_2 - \lambda_1} = A_2^0 - A_1^0 \frac{C_2}{C_1}\frac{\lambda_2}{\lambda_2 - \lambda_1}$$

where the A^0 are the true initial activities.

Lewis Research Center
National Aeronautics and Space Administration
Cleveland, Ohio, July 25, 1962

APPENDIX - MULTIPLE DECAY ANALYSIS PROGRAM

```
C     MULTIPLE DECAY ANALYSIS PROGRAM
      DIMENSION ARRAY(20, 20), PHI(400), TAU(400), WT(400), COL(20),
     1ANS(20),  UNIT(21), A(10), B(10), OUT(60), NOUT(60), RESID(400),
     2CALC(400), WTRES(400)
      SENSE LIGHT 0
  100 WRITE OUTPUT TAPE 6, 10
   10 FORMAT(46H1    UNFOLDING SOLUTION OF MULTIPLE DECAY DATA   )
C         INPUT BEGINS
      READ INPUT TAPE 7, 20
   20 FORMAT(72H     DATA OF 3/18/60. (TIMES MEASURED IN MINUTES)
     1                )
      READ INPUT TAPE 7, 30, N, M, NC, NE
   30 FORMAT(22H0    NO. OF COMPONENTS= I2, 21H, NO. OF DATA POINTS= I3,
     123H, NO. FIXED COMPONENTS=I1 / 25H NO. OF FIXED HALF LIVES= I1)
      READ INPUT TAPE 7, 40, (TAU(I), PHI(I), WT(I), I=1, M)
   40 FORMAT(6H TIME= F8.3, 9H    COUNT= F8.3, 10H    WEIGHT= F8.3)
      READ INPUT TAPE 7, 50, (A(I), B(I), I=1, N)
   50 FORMAT(9H        A= F8.3, 5H    B= F8.3)
      READ INPUT TAPE 7, 60, CONV, WTFACT
   60 FORMAT(27H    CONVERGENCE REQUIREMENT= F8.5, 15H WEIGHT FACTOR=
     1F10.5 )
      WRITE OUTPUT TAPE 6, 20
      WRITE OUTPUT TAPE 6, 30, N, M, NC, NE
C         COMPUTATION BEGINS,  COMPUTE WEIGHTS
      DO 140 I=1, M
      WT(I)= WT(I)*WTFACT
      IF(WT(I)) 110, 120, 130
  110 WT(I)=-WT(I)
      GO TO 140
  120 WT(I)=WTFACT/PHI(I)
      GO TO 140
  130 WT(I)=WT(I)/PHI(I)**2
  140 CONTINUE
      DO 150 I=1, N
  150 B(I)=-0.69315/B(I)
C         GENERATE INDICES
      NJJ=2*N
      NPNC=N
      N=N-NC
      NPO=N+1
      NJ=2*N-NE
      NI= 2*N+1
      NL=N-NE
      NK=NL+1
C         INTERMEDIATE OUTPUT
      WRITE OUTPUT TAPE 6, 65, (I, I=1, NPNC)
   65 FORMAT(40H0            TRIAL VALUES OF A(I) AND B(I)   / 7H0    I= I5,
     19I11 )
  250 DO 255 I=1, NPNC
      J=I+NPNC
      OUT(I)=A(I)
  255 OUT(J)=-0.69315/B(I)
      DO 260 I=1, NJJ
      CALL TEFOUT(OUT(I), NOUT(I))
  260 CONTINUE
      WRITE OUTPUT TAPE 6, 70, (OUT(I), NOUT(I), I=1, NPNC)
   70 FORMAT(6H0A(I)= 10(F8.3, I3))
      II=NPNC+1
      WRITE OUTPUT TAPE 6, 75, (OUT(I), NOUT(I), I=II, NJJ)
   75 FORMAT(6H B(I)= 10(F8.3, I3))
```

```
C          COMPUTATION PROCEEDS, CALCULATE PSI(I)
  160 DO 170 I=1, NJ
      COL(I)=0.0
      DO 170 J=1, NJ
  170 ARRAY(I,J)=0.0
      DO 200 J=1, M
      DO 180 K=1, NJ
  180 UNIT(K)=0.0
      UNIT(NI)=PHI(J)
      DO 190 I=1, N
      Y=B(I)*TAU(J)
      IF(25.0+Y) 183, 186, 186
  183 BOLD=0.0
      GO TO 190
  186 BOLD=EXPF(Y)
      UNIT(I)=A(I)*BOLD
      K=I+N
      UNIT(K)=-A(I)*B(I)*TAU(J)*BOLD
  190 UNIT(NI)=UNIT(NI)-UNIT(I)
C          CORRECTION FOR FIXED COMPONENTS
      IF(NC) 191, 191, 1910
 1910 DO 1930 I=NPO, NPNC
      Y=B(I)*TAU(J)
      IF(25.0+Y) 1930, 1920, 1920
 1920 UNIT(NI)=UNIT(NI)-A(I)*EXPF(Y)
 1930 CONTINUE
C          SET UP MATRIX AND COLUMN VECTOR
  191 DO 200  L=1, NJ
      DO 199 K=1, NJ
  199 ARRAY(K,L)=ARRAY(K,L)+WT(J)*UNIT(K)*UNIT(L)
  200 COL(L)=COL(L)+UNIT(L)*UNIT(NI)*WT(J)
C          INVERT MATRIX
  211 CALL MATINV(ARRAY, NJ)
      IF  (SENSE LIGHT 1) 212, 213
  212 WRITE OUTPUT TAPE 6, 2120
 2120 FORMAT(20H0 RESTART BY MATINV   )
      GO TO 100
C         COMPUTE CORRECTION TERMS
  213 DO 220 I=1, NJ
      ANS(I)=0.0
      DO 220 J=1, NJ
C          CORRECT EXPONENTS
  220 ANS(I)=ANS(I)+ARRAY(I,J)*COL(J)
      IF(NL)  226, 226, 221
  221 DO 225 I=1,NL
      II= I+N
  225 B(I)=B(I)*(1.0-ANS(II))
  226 XMAX= 0.0
C          CORRECT COEFFICIENTS
      DO 230 I=1,N
      A(I)=A(I)*(1.0+ANS(I))
  230 XMAX=MAX1F(XMAX, ABSF(ANS(I)))
C          CONVERGENCE TEST
      IF (XMAX-CONV) 240, 240, 250
C          COMPUTE CALCULATED VALUES
  240 DO 245 I=1, M
      CALC(I)=0.0
      DO 245  J=1, NPNC
  245 CALC(I)=CALC(I)+A(J)*EXPF(B(J)*TAU(I))
      SQRES=0.0
```

```
C        COMPUTE PERCENTAGE DEVIATION AND WEIGHTED SQUARED RESIDUALS
      DO 246  I=1, M
      RESID(I)=(CALC(I)-PHI(I))/PHI(I)
      WTRES(I)= WT(I)*(RESID(I)*PHI(I))**2
  246 SQRES= SQRES+WTRES(I)
C        COMPUTE WEIGHTED VARIANCE
      WVAR=SQRES/FLOATF(M-NJ-1)
C        OUTPUT BEGINS, DEFINE OUTPUT QUANTITIES
  280 DO 281 I=1,N
      J=6*I-5
      OUT(J)= A(I)
      OUT(J+1)= -0.69315/B(I)
      OUT(J+2)=ANS(I)
  281 OUT(J+4)=A(I)*SQRTF(ARRAY(I,I)*WVAR)
C        BRANCH IF ALL FIXED EXPONENTS
      IF(NL) 282, 282, 284
  282 DO 283 I=1,N
      J=6*I-5
      OUT(J+3)=0.0
  283 OUT(J+5)=0.0
      GO TO 291
  284 DO 285 I=1,NL
      J=6*I-5
      II= I+N
      OUT(J+3)=-ANS(II)
  285 OUT(J+5)=(-0.69315/B(I))*SQRTF(ARRAY(II,II)*WVAR)
C        BRANCH IF ANY FIXED EXPONENTS
      IF(NE)  291, 291, 286
  286 DO 290 I=NK,N
      J=6*I-5
      OUT(J+3)=0.0
  290 OUT(J+5)=0.0
  291 J=6N
C        OUTPUT COEFFICIENTS AND ERRORS
      DO 300 I=1, J
      CALL TEFOUT(OUT(I), NOUT(I))
  300 CONTINUE
      WRITE OUTPUT TAPE 6, 90, (OUT(I), NOUT(I), I=1, J)
   90 FORMAT(24H1 FINAL VALUES OBTAINED // 94H       A                B
     1        ALPHA           BETA            SIGMA(A)         SIGMA(B)
     2     //(6(F11.4, I3)))
C        BRANCH IF ANY FIXED COMPONENTS
      IF(NC) 355, 355, 340
  340 DO 350 I=NPO, NPNC
      OUT(1)=A(I)
      CALL TEFOUT(OUT(1), NOUT(1))
      OUT(2)=-0.69315/B(I)
      CALL TEFOUT (OUT(2), NOUT(2))
      WRITE OUTPUT TAPE 6, 95, OUT(1), NOUT(1), OUT(2), NOUT(2)
   95 FORMAT(2(F11.4, I3))
  350 CONTINUE
C        OUTPUT WEIGHTED VARIANCE AND WEIGHTED SQUARED RESIDUALS
  355 OUT(1)=WVAR
      CALL TEFOUT(OUT(1), NOUT(1))
      OUT(2)=SQRES
      CALL TEFOUT(OUT(2), NOUT(2))
      WRITE OUTPUT TAPE 6, 975, OUT(1), NOUT(1), OUT(2),  NOUT(2)
  975 FORMAT(19HOWEIGHTED VARIANCE=F11.4,I3,40H       SUM OF WEIGHTED SQU
     1ARED RESIDUALS=F11.4, I3)
C        OUTPUT DATA POINT QUANTITIES
```

```
360 WRITE OUTPUT TAPE 6, 40
400 FORMAT(114H0     TIME          OBSERVED RATE       WEIGHT USED       C
   1ALCULATED RATE       PER CENT DEVIATION    WT SQ RESIDUALS  / 2H   )
    DO 330 I= 1,M
    OUT(1)=TAU(I)
    OUT(2)=PHI(I)
    OUT(3)=WT(I)
    OUT(4)=CALC(I)
    OUT(5)=100.0*RESID(I)
    OUT(6)= WTRES(I)
    DO 320 J=1, 6
    CALL TEFOUT(OUT(J), NOUT(J))
320 CONTINUE
330 WRITE OUTPUT TAPE 6, 410, (OUT(J), NOUT(J), J=1, 6)
410 FORMAT(F9.4,I3, F13.4,I3, F15.4,I3, F16.4,I3, F21.4,I3, F17.4,I3)
    GO TO 100


C     THIS SUBROUTINE INVERTS A SQUARE MATRIX WITH SIDE M IN SIZE.
      SUBROUTINE MATINV(X,M)
      DIMENSION X(20,20), K(20)
      FREQUENCY 105(1,0,1), 125(5,1,5), 155(1,0,1), 160(1,0,0), 175(1,0,
     11), 230(1,2,1)
C         TEST FOR MATRIX SIZE
      IF (M-1)  190, 20, 30
   20 X(1,1)= 1.0/X(1,1)
      GO TO 245
C         INITIALIZE ROUTINE CONSTANTS
   30 TEST=0.0
      MM=M-1
      DO 100 I=1, M
  100 K(I)=I
      DO 150 N=1, M
C         TEST PIVOTAL ELEMENT
  105 IF(X(1,N)) 110, 160, 11
C         MATRIX INVERSION BY ONE COLUMN
  110 A=X(1,N)
      DO 120 I=1, MM
  120 X(I,N)=X(I+1,N)/A
      X(M,N)=1.0/A
      DO 150 J=1, M
  125 IF(J-N) 130, 150, 130
  130 A=X(1,J)
      DO 140 I=1, MM
  140 X(I,J)=X(I+1,J)-A*X(I,N)
      X(M,J)=-A*X(M,N)
  150 CONTINUE
C         SEARCH FOR ROW WITH NON-ZERO ELEMENT IN FIRST COLUMN
  155 IF(TEST) 220, 245, 220
  160 IF(N-M) 170, 190, 190
  170 NN=N+1
      DO 180 L=NN, M
  175 IF(X(1,L)) 200, 180, 20
  180 CONTINUE
C         SENSE LIGHT1 EXIT IF MATRIX HAS NO INVERSE
  190 SENSE LIGHT 1
      GO TO 245
C         REORDER MATRIX TO OBTAIN NON-ZERO PIVOTAL ELEMENT
  200 DO 210 I=1, M
      HOLD=X(I,N)
      X(I,N)=X(I,L)
```

```
  210 X(I,L)=HOLD
      LL=K(N)
      K(N)=K(L)
      K(L)=LL
      TEST=1.0
      GO TO 110
C         UNSCRAMBLE REORDED MATRIX TO OBTAIN CORRECT INVERSE
  220 DO 240 I=1,MM
  230 IF(K(I)-I) 250, 240, 25
  240 CONTINUE
  245 RETURN
  250 L=K(I)
      DO 260 J=1, M
      HOLD=X(I,J)
      X(I,J)=X(L,J)
  260 X(L,J)=HOLD
      K(I)=K(L)
      K(L)=L
      GO TO 230

      SUBROUTINE TEFOUT(X,I)
      A=40.0+(0.43429448*LOGF(ABSF(X)))
      X=SIGNF(EXPF(MODF(A,1.0)/0.43429448),X)
      I=XINTF(A)-40
      RETURN
      END
```

REFERENCES

1. Harmer, Don S.: A Univac Program for the Analysis of Radioactive Decay Curves. BNL 544, Brookhaven Nat. Lab., Mar. 1959.

2. Moore, R. H., and Zeigler, R. K.: The Solution of the General Least Squares Problem with Special Reference to High-Speed Computers. LA-2367, Los Alamos Sci. Lab., Mar. 1960.

3. Hald, A.: Statistical Theory with Engineering Application. John Wiley & Sons, Inc., 1952, p. 552.

4. Hildebrand, F. B.: Introduction to Numerical Analysis. Mc-Graw-Hill Book Co., Inc., 1956, p. 264-268.

5. Scarborough, James B.: Numerical Mathematical Analysis. Third ed., The Johns Hopkins Press, 1955, pp. 519-531.

6. Cziffra, Peter, and Moravcsik, Michael J.: A Practical Guide to the Method of Least Squares. UCRL 8523, Lawrence Radiation Lab., Univ. Calif., June 5, 1959.

TABLE I. - DATA USED TO TEST PROGRAM

```
0       MANUFACTURED SAMPLE
0     NO. OF COMPONENTS= 3, NO. OF DATA POINTS= 20, NO. FIXED COMPONENTS=0
  NO OF FIXED HALF LIVES=1
 TIME=    5       COUNT=  17689     WEIGHT=530
 TIME=   10       COUNT=  16340     WEIGHT=491
 TIME=   15       COUNT=  15246     WEIGHT=455
 TIME=   20       COUNT=  14198     WEIGHT=424
 TIME=   40       COUNT=  10815     WEIGHT=649
 TIME=   60       COUNT=   8581     WEIGHT=516
 TIME=   80       COUNT=   7072     WEIGHT=423
 TIME=  100       COUNT=   6014     WEIGHT=358
 TIME=  120       COUNT=   5150     WEIGHT=308
 TIME=  150       COUNT=   4240     WEIGHT=383
 TIME=  180       COUNT=   3598     WEIGHT=326
 TIME=  210       COUNT=   3162     WEIGHT=284
 TIME=  240       COUNT=   2780     WEIGHT=251
 TIME=  300       COUNT=   2280     WEIGHT=409
 TIME=  360       COUNT=   1924     WEIGHT=348
 TIME=  420       COUNT=   1698     WEIGHT=307
 TIME=  540       COUNT=   1444     WEIGHT=524
 TIME=  660       COUNT=   1346     WEIGHT=483
 TIME=  780       COUNT=   1288     WEIGHT=465
 TIME=  900       COUNT=   1271     WEIGHT=457
        A=  125.1     B=     30.0
        A=   90.0     B=    110.0
        A=   10.0     B= 99999.9
   CONVERGENCE REQUIREMENT= 0.001    WEIGHT FACTOR= 1.0
```

TABLE II. - SOLUTION OF MULTIPLE DECAY DATA

[1.234 -1 denotes 1.234×10^{-1}, etc.]

UNFOLDING SOLUTION OF MULTIPLE DECAY DATA

MANUFACTURED SAMPLE

NO. OF COMPONENTS= 3, NO. OF DATA POINTS= 20, NO. FIXED COMPONENTS=0
NO OF FIXED HALF LIVES=1

TRIAL VALUES OF A(I) AND B(I)

I=	1	2	3
A(I)=	1.251 2	9.000 1	1.000 1
B(I)=	3.000 1	1.100 2	10.000 4
A(I)=	1.027 2	7.630 1	1.261 1
B(I)=	2.800 1	1.039 2	10.000 4
A(I)=	1.029 2	7.627 1	1.259 1
B(I)=	2.752 1	1.025 2	10.000 4

FINAL VALUES OBTAINED

A	B	ALPHA	BETA	SIGMA(A)	SIGMA(B)
1.0291 2	2.7521 1	2.2271 -4	-1.4330 -4	1.4154 0	5.3332 -1
7.6252 1	1.0247 2	-2.8609 -4	4.3217 -5	1.5463 0	1.2449 0
1.2587 1	10.0000 4	-9.7491 -5	0. 0	4.3858 -2	0. 0

WEIGHTED VARIANCE= 8.9494 -1 SUM OF WEIGHTED SQUARED RESIDUALS= 1.2529 1

TIME	OBSERVED RATE	WEIGHT USED	CALCULATED RATE	PER CENT DEVIATION	WT SQ RESIDUALS
5.0000 0	1.7689 2	1.6938 0	1.7704 2	8.3583 -2	3.7026 -2
1.0000 1	1.6340 2	1.8390 0	1.6385 2	2.7524 -1	3.7196 -1
1.5000 1	1.5246 2	1.9575 0	1.5201 2	-2.9287 -1	3.9026 -1
2.0000 1	1.4198 2	2.1033 0	1.4138 2	-4.2534 -1	7.6709 -1
4.0000 1	1.0815 2	5.5487 0	1.0834 2	1.7346 -1	1.9527 -1
6.0000 1	8.5810 1	7.0077 0	8.6104 1	3.4241 -1	6.0500 -1
8.0000 1	7.0720 1	8.4578 0	7.0686 1	-4.7809 -2	9.6686 -3
1.0000 2	6.0140 1	9.8982 0	5.9638 1	-8.3450 -1	2.4931 0
1.2000 2	5.1500 1	1.1613 1	5.1450 1	-9.7993 -2	2.9576 -2
1.5000 2	4.2400 1	2.1304 1	4.2571 1	4.0223 -1	6.1965 -1
1.8000 2	3.5980 1	2.5182 1	3.6243 1	7.3022 -1	1.7383 0
2.1000 2	3.1620 1	2.8405 1	3.1509 1	-3.5033 -1	3.4855 -1
2.4000 2	2.7800 1	3.2478 1	2.7848 1	1.7203 -1	7.4285 -2
3.0000 2	2.2800 1	7.8678 1	2.2636 1	-7.2018 -1	2.1213 0
3.6000 2	1.9240 1	9.4009 1	1.9246 1	2.8665 -2	2.8594 -3
4.2000 2	1.6980 1	1.0648 2	1.7003 1	1.3658 -1	5.7264 -2
5.4000 2	1.4440 1	2.5130 2	1.4516 1	5.2815 -1	1.4617 0
6.6000 2	1.3460 1	2.6660 2	1.3407 1	-3.9343 -1	7.4762 -1
7.8000 2	1.2880 1	2.8030 2	1.2909 1	2.2293 -1	2.3109 -1
9.0000 2	1.2710 1	2.8289 2	1.2682 1	-2.2321 -1	2.2769 -1

TABLE III. - DISTRIBUTION OF F: VALUES OF $F_a(\mathscr{M})$ FOR GIVEN $\mathscr{M}$ AND a

a	Number of degrees of freedom, $\mathscr{M}$								
	5	10	15	20	30	40	60	120	∞
0.999	1.73×10^{-6}	1.65×10^{-6}	1.62×10^{-6}	1.61×10^{-6}	1.60×10^{-6}	1.59×10^{-6}	1.58×10^{-6}	1.58×10^{-6}	1.57×10^{-6}
.99	1.75×10^{-6}	1.65×10^{-4}	1.62×10^{-4}	1.61×10^{-4}	1.60×10^{-4}	1.59×10^{-4}	1.58×10^{-4}	1.58×10^{-4}	1.57×10^{-4}
.95	4.35×10^{-3}	4.13×10^{-3}	4.07×10^{-3}	4.03×10^{-3}	4.00×10^{-3}	3.98×10^{-3}	3.96×10^{-3}	3.95×10^{-3}	3.93×10^{-3}
.90	.0175	.0166	.0163	.0162	.0160	.0160	.0159	.0158	.0158
.80	.0713	.0676	.0666	.0660	.0655	.0650	.0645	.0645	.0640
.75	.113	.107	.105	.104	.103	.103	.1025	.102	.1015
.50	.523	.490	.477	.472	.466	.464	.461	.458	.454
.25	1.69	1.49	1.43	1.40	1.38	1.36	1.35	1.34	1.32
.10	4.06	3.29	3.07	2.97	2.88	2.84	2.79	2.75	2.71
.05	6.61	4.96	4.54	4.35	4.17	4.08	4.00	3.92	3.84
.01	16.26	10.04	8.68	8.10	7.56	7.31	7.08	6.35	6.63
.005	22.78	12.83	10.80	9.94	9.18	8.83	8.49	8.18	7.88
.001	47.18	21.04	16.59	14.82	13.29	12.61	11.97	11.38	10.38

V62 16298

ERRATA

NASA TECHNICAL NOTE D-1453

A FORTRAN II PROGRAM FOR ANALYSIS OF

RADIOACTIVE DECAY CURVES

By John L. Need and Theodore E. Fessler

October, 1962

Page 7: Paragraph 2 should read

It is $(m - 2n - 1)Q_n$ that possesses a χ^2 distribution with $(m - 2n - 1)$ degrees of freedom. Sometimes the minimum value obtainable from a set of data will be considerably different from 1, in which case it may be concluded that either the basic assumptions behind equation (1) do not apply or that there is some internal inconsistency in the data. It will be noticed that the statistic

$$Q_n = \frac{1}{m - 2n - 1} \sum_{k=1}^{m} \frac{\tau_k^2}{N_k} \left[\varphi(t_k) - \sum_{i=1}^{n} \psi_i(t_k) \right]^2$$

$$= \frac{1}{m - 2n - 1} \sum_{k=1}^{m} \frac{\left[N_k - \sum_{i=1}^{n} \tau_k \psi_i(t_k) \right]^2}{N_k}$$

This is of the form

$$\frac{1}{m - 2n - 1} \sum_{\substack{\text{all} \\ \text{data} \\ \text{points}}} \frac{(\text{observed count - expected count})^2}{\text{observed count}}$$

and is, therefore, not exactly $\chi^2/(m - 2n - 1)$.

Page 17: Statement 291 should read

```
291 J = 6*N
```

Page 19: Subroutine TEFOUT should read

```
      SUBROUTINE TEFOUT (X,I)
      IF (X)2,1,2
    1 I=0
      RETURN
    2 A=40.0+(0.43429448*LOGF(ABSF(X)))
      X=SIGNF(EXPF(MODF(A,1.0)/0.43429448),X)
      I=XINTF(A)-40
      RETURN
      END
```


NASA-Langley, 1963 E-1620

NASA TN D-1453
National Aeronautics and Space Administration.
A FORTRAN II PROGRAM FOR ANALYSIS OF RADIOACTIVE DECAY CURVES. John L. Need and Theodore E. Fessler. October 1962. 23p. OTS price, $0.75. (NASA TECHNICAL NOTE D-1453)

The program can be used to find the half-lives and the initial magnitudes of the components in a measured decay curve. Any of the components can be fixed as to magnitude and half-life and any of the half-lives can be fixed, provided that there be at least one variable remaining. The calculation proceeds by an iterative least-squares fit to the experimental data. Convergence is not guaranteed, and a stepwise approach may be necessary. The output consists of the amplitudes and half-lives of all the components together with the best estimate of the errors in those magnitudes calculated by the program. Information is also given that permits evaluation of the goodness of fit. The program can handle up to 10 components (including background) and up to 400 observations.

I. Need, John L.
II. Fessler, Theodore E.
III. NASA TN D-1453

(Initial NASA distribution: 31, Physics, nuclear and particle.)

NASA

NASA TN D-1453
National Aeronautics and Space Administration.
A FORTRAN II PROGRAM FOR ANALYSIS OF RADIOACTIVE DECAY CURVES. John L. Need and Theodore E. Fessler. October 1962. 23p. OTS price, $0.75. (NASA TECHNICAL NOTE D-1453)

The program can be used to find the half-lives and the initial magnitudes of the components in a measured decay curve. Any of the components can be fixed as to magnitude and half-life and any of the half-lives can be fixed, provided that there be at least one variable remaining. The calculation proceeds by an iterative least-squares fit to the experimental data. Convergence is not guaranteed, and a stepwise approach may be necessary. The output consists of the amplitudes and half-lives of all the components together with the best estimate of the errors in those magnitudes calculated by the program. Information is also given that permits evaluation of the goodness of fit. The program can handle up to 10 components (including background) and up to 400 observations.

I. Need, John L.
II. Fessler, Theodore E.
III. NASA TN D-1453

(Initial NASA distribution: 31, Physics, nuclear and particle.)

NASA

NASA TN D-1453
National Aeronautics and Space Administration.
A FORTRAN II PROGRAM FOR ANALYSIS OF RADIOACTIVE DECAY CURVES. John L. Need and Theodore E. Fessler. October 1962. 23p. OTS price, $0.75. (NASA TECHNICAL NOTE D-1453)

The program can be used to find the half-lives and the initial magnitudes of the components in a measured decay curve. Any of the components can be fixed as to magnitude and half-life and any of the half-lives can be fixed, provided that there be at least one variable remaining. The calculation proceeds by an iterative least-squares fit to the experimental data. Convergence is not guaranteed, and a stepwise approach may be necessary. The output consists of the amplitudes and half-lives of all the components together with the best estimate of the errors in those magnitudes calculated by the program. Information is also given that permits evaluation of the goodness of fit. The program can handle up to 10 components (including background) and up to 400 observations.

I. Need, John L.
II. Fessler, Theodore E.
III. NASA TN D-1453

(Initial NASA distribution: 31, Physics, nuclear and particle.)

NASA

NASA TN D-1453
National Aeronautics and Space Administration.
A FORTRAN II PROGRAM FOR ANALYSIS OF RADIOACTIVE DECAY CURVES. John L. Need and Theodore E. Fessler. October 1962. 23p. OTS price, $0.75. (NASA TECHNICAL NOTE D-1453)

The program can be used to find the half-lives and the initial magnitudes of the components in a measured decay curve. Any of the components can be fixed as to magnitude and half-life and any of the half-lives can be fixed, provided that there be at least one variable remaining. The calculation proceeds by an iterative least-squares fit to the experimental data. Convergence is not guaranteed, and a stepwise approach may be necessary. The output consists of the amplitudes and half-lives of all the components together with the best estimate of the errors in those magnitudes calculated by the program. Information is also given that permits evaluation of the goodness of fit. The program can handle up to 10 components (including background) and up to 400 observations.

I. Need, John L.
II. Fessler, Theodore E.
III. NASA TN D-1453

(Initial NASA distribution: 31, Physics, nuclear and particle.)

NASA